
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2017-06_D17wk51’.

3. Corrections to E-UTRA MAC test case 7.1.3.9
Change 1 – Correction to function ‘f_TC_7_1_3_9_EUTRA’
	Function name
	f_TC_7_1_3_9_EUTRA ()

	Reason for change
	1.    At step 2 and step 8, a PDCP SDU size of 38 Bytes (= 304 bits) is chosen, which would fit into a TBS size of 328 bits suitable for eMTC (as per the TS 36.523-3 TBS sheets).
But TTCN expects the same DRB data. This leads to a TC failure when the UE loops back the received data with 4 bytes of additonal padding.
2.    Corrupted implementation of “R5s170939_MCC160Comments1.zip”, maybe because an additional change was provided shortly before the ATS_17wk51 delivery deadline. 

	Summary of change
	1.    At step 2 and step 8, a PDCP SDU of size 36 Bytes was chosen instead because this fits incl. 4 Bytes of padding into the TBS size of 328 bits that is suitable for eMTC as per 36.523-3 TBS sheet.

2.    SFN has to be incremented for the HARQ ACK / NACK, i.e. before step 6 and not before step 3.

	TTCN module
	LTE\7_1\MAC_713.ttcn

	MCC160 Comment
	1. The reason for change is incorrect. The problem can be that with SS configured in DL MAC transparent mode, a final DL MAC PDU of 344 bits is provided from TTCN. This size of MAC TBS of 344 is valid in legacy DCI format 1A TBS sheets but not possible for CAT M1 UE’s with DCI 6-1A. Hence accepted conditional to prose CR agreement in RAN5.
Also a new PDCP SDU template created to be used in step 8, to be different from step 2.

template (value) PDCP_SDU_Type crs_PDCP_SDU_L36_1 := f_GetN_OctetsFromPRBS(550, 36);

2. Accepted. Implemented before step 3, but after step 5 transmission of RRC Connection Reconfiguration.


Before  change
…

  /*
   * @desc      REFERENCE TS 36.523-1 clause 7.1.3.9
   * @status    APPROVED (LTE)
   */
  function f_TC_7_1_3_9_EUTRA ( ) runs on EUTRA_PTC
  { /* L2/MAC
     * 7.1.3.9 : MAC reset-DL
     */
    const HarqProcessId_Type tsc_HARQ_Process := 3; // selected the highest allowed HARQ process in TDD config 0
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing2;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var SYSTEM_IND v_ReceivedSYSIND;
    var float v_HarqIndWindowSize;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    //@sic R5s110576 sic@
    var template (value) RadioBearerList_Type v_DRB :={cs_OneDRB_ConfigAM(tsc_DRB1, cds_PDCP_Config_DRB_AM_NoPDCP_StatusReport, //@sic R5s110687 sic@
                                                                          cs_RLC_Configuration_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)))};
    var template (value) DRB_ToAddModList v_DRB_ToAddModList := {cs_508_DRB_ToAddMod_Common(tsc_DRB1,
                                                                                            cds_PDCP_Config_DRB_AM_NoPDCP_StatusReport,//@sic R5s110687 sic@
                                                                                            cs_508_RLC_Config_DRB_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)),
                                                                                            cs_508_LogicalChannelConfig_DRB_AM)};
    var octetstring v_EncodedPdcpPdu;
    var octetstring v_EncodedRlcPdu;
    var SubFrameNumber_Type v_SubFrameRRCReconfigAck := 8; //@sic R5s170939 sic@
    timer t_Watchdog:=5.0; // General purpose wait timer
    if (pc_ue_CategoryDL_M1)  //@sic R5s170939 sic@
    {
    v_SubFrameRRCReconfigAck := 0;
    }
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    // Config Cell 1 for 1 AM DRB
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_SetCellPower(eutra_Cell1, -85);
    //@sic R5s100456, cell config moved above set attenuation  sic@
    // bring up cell 2 with T0 setting
    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);
    f_EUTRA_CellConfig_Def (eutra_Cell2);
    f_SS_ConfigDLHARQProcess_Subset(eutra_Cell1, -, cs_HarqProcessList_012);// To allow only first 3 HARQ processes for autonomous SS selection
    f_SS_ConfigDLHARQProcess_Subset(eutra_Cell2, -, cs_HarqProcessList_012);// To allow only first 3 HARQ processes for autonomous SS selection
    // Bring UE to  state 4
    f_EUTRA_Preamble ( eutra_Cell1,STATE2A_TESTLOOP_ModeA );
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    //@sic R5s110576 sic@
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell1, v_DRB);
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling(false, v_DRB_ToAddModList); //  p_PDCP_StatusReportRequired is false
    f_Delay(0.660); // SS is not stopped for TA transmission to prevent UE from entering DRX
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);
    // preconfig for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell2, tsc_C_RNTI_Def2);     /* @sic acc. to R5s100029 change 3 R5s120629 sic@ */
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    //@sic R5s150413 sic@
    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell2, cs_OneDRB_ConfigAM);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2); // @sic R5-101050 sic@
    //@siclog "Step 1" siclog@
    // To stop UL Grant Transmision
    //@sic R5s110201 sic@
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1,cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    //@sic R5s100852r2 MCC change sic@
    // to configure SS to report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // 38 bytes PDCP SDU +2 byte PDCP header
    // 2 bytes RLC Header +2 byte MAC Header =352 bits, nearest TB  size =376 => padding =24 bits
    // 1 byte padding header and 2 bits padding
    //@siclog "Step 1" siclog@
    // Sequence is changed to facilitate SS implementations.
    f_EUTRA_SetCellPower (eutra_Cell2, -79);
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 400);// schedule 400 ms in future
    // set CRC error as error mode
    f_SS_ConfigCRC_ErrorMode(eutra_Cell1, cs_DL_SCH_CRC_C_RNTI_Error(Erroneous), cs_TimingInfo(v_Timing));
    v_Timing.Subframe.Number := 0; // suitable for both FDD and TDD
    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 30);// schedule 20-30 ms in future
    //@siclog "Step 2" siclog@
    //@sic R5s120629 sic@
    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_L38)));
    v_EncodedRlcPdu   := f_RLC_AMD_1PDU_Encvalue(0, tsc_NoPoll, v_EncodedPdcpPdu);
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo(v_Timing), //cs_TimingInfo_Now,
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,
                                                                                             v_EncodedRlcPdu)),
                                          cs_HarqProcessAssignment_Specific(tsc_HARQ_Process)));
    // set CRC error as normal mode 60 ms after the scheduled transmission
    // so that 5 transmissions are performed.
    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);
    f_SS_ConfigCRC_ErrorMode(eutra_Cell1, cs_DL_SCH_CRC_C_RNTI_Error(Normal), cs_TimingInfo( v_Timing ));
    //@sic R5s120629 sic@
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1,cs_MAC_TestMode_NormalMode,-,cs_TimingInfo( v_Timing ));
    //@siclog "Step 5" siclog@
    // Tell the SS about the HO
    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverInit_REQ(eutra_Cell1, eutra_Cell2)); // @sic R5s100556 sic@
    v_Auth_Params := f_EUTRA_Security_Get();
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell2, v_Auth_Params, 0);
    f_EUTRA_Security_Set ( v_Auth_Params );
    f_EUTRA_SS_ConfigRachProcedure_HO(eutra_Cell2, cs_TimingInfo_Now, tsc_C_RNTI_Def2);
    f_EUTRA_CellInfo_SetC_RNTI (eutra_Cell2, tsc_C_RNTI_Def2);
    // stop periodic TA on source cell
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_None)); // Note: since the UE is configured for TA infinitive it does not matter that SS reconfiguration is not scheduled
    //@sic R5s150413 sic@
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell2, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    // configure UL grant configuration on target cell    @sic R5s100515 sic@
    f_EUTRA_ULGrantConfiguration_Common(eutra_Cell2, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def, cs_UplinkTimeAlignment_Synch_None, cs_UL_GrantConfig_Def);
    v_Timing2 := f_EUTRA_NB_TimingInfoAdd(v_Timing, 10);
    if (pc_ue_CategoryDL_M1)  //@sic R5s170939 sic@
    {
    v_SubFrameRRCReconfigAck := 0;
    v_Timing2.SFN.Number := ((v_Timing2.SFN.Number) + 1) mod 1024;
    }
    //@sic R5s100240 sic@
    f_EUTRA_508RRC_ConnReconfigHO_IntraLTE (eutra_Cell1,
                                            eutra_Cell2,
                                            cs_TimingInfo_SubFrameOffset( v_Timing2.SFN.Number, 4, v_Timing2.HSFN.Number )); // next sub frame suitable for both FDD and TDD @sic R5w160204r1 sic@
    //@siclog "Step 3" siclog@
    t_Watchdog.start;
    SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack ));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    t_Watchdog.stop;
    //@siclog "Step 6" siclog@
    // Check: does the UE transmit a RRCConnectionReconfigurationComplete message to confirm the reconfiguration of the radio resources?
    t_Watchdog.start;
    alt {
      [] SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) ) -> value v_SRB_COMMON_IND
        {
          v_Timing := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; //store TimeStamp when RRCConnectionReconfiguration has been received
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          t_Watchdog.stop;
        }
      []  SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack ))
        {
          repeat;
          // any NACKS received
        }
      //@sic R5s170939 sic@
      []  SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, cr_TimingInfo( v_Timing2.SFN.Number, v_SubFrameRRCReconfigAck ), ack )) // subframe 8 or 0 for ACK both for FDD and TDD configuration 1
        //note: retransmissions not allowed in this test case
        {
          repeat;
          // ACK  received for RRC connection reconfiguration
        }
    };
    //@siclog "Step 7" siclog@
    t_Watchdog.start (0.5); // watch dog timer to check 100ms as per prose which is less than 500ms, actual 100ms is checked based on time stamps
    alt {
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack )) -> value v_ReceivedSYSIND
        {
          //Check if the received HARQ Ind is within a window of 100 ms
          v_Timing2 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
          v_HarqIndWindowSize := f_EUTRA_SetTimerToleranceMin(eutra_Cell1, l2Timer, 0.1); //100 ms to wait safely no SR Rx
          if (f_EUTRA_SubFrameTimingDuration_Float(v_Timing, v_Timing2) < v_HarqIndWindowSize) {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7: UE is sending HARQ NACK in cell 1");
          }
        }
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell2, ?, nack )) -> value v_ReceivedSYSIND
        {
          //Check if the received HARQ Ind is within a window of 100 ms
          v_Timing2 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
          v_HarqIndWindowSize := f_EUTRA_SetTimerToleranceMin(eutra_Cell1, l2Timer, 0.1); //100 ms to wait safely no SR Rx
          if (f_EUTRA_SubFrameTimingDuration_Float(v_Timing, v_Timing2) < v_HarqIndWindowSize) {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7: UE is sending HARQ NACK in cell 2");
          }
        }
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell2, ?, ack ))      // HARQ ACK for Status PDU sent by SS RLC for RLC reconfiguration complete
        {
          repeat;
        }
      []  t_Watchdog.timeout
        {
        }
    }
    // disable harq reporting in cell 2
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);
    // default grant config
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell2);
    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverComplete_REQ(eutra_Cell2)); // @sic R5s100556, R5s100560 sic@
    // configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell2);     /* @sic acc. to R5s100029 change 3 sic@ */
    //@sic R5s150992 sic@
    f_EUTRA_SS_SRBs_DRBs_Reset(eutra_Cell1);
    // to enable reception of SR in cell 2
    //@sic R5s110456 sic@
    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);
    //@siclog "Step 8" siclog@
    // SS guarantees NDI to be same as step2, as
    //"The initial value of the NDI shall be the same for all HARQ processes and cells".
    //@sic R5s120629 sic@
    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_L38_1)));
    v_EncodedRlcPdu   := f_RLC_AMD_1PDU_Encvalue(0, tsc_NoPoll, v_EncodedPdcpPdu);
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell2,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo_Now,
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,
                                                                                             v_EncodedRlcPdu)),
                                         cs_HarqProcessAssignment_Specific(tsc_HARQ_Process)));
    //@siclog "Step 9" siclog@
    t_Watchdog.start;
    SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell2, ?));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9: UE sent SR in cell 2");
    t_Watchdog.stop;
    //@siclog "Step 10" siclog@
    // Automatically taken care of
    //@siclog "Step 11" siclog@
    t_Watchdog.start;
    //@sic R5s120629 sic@
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (eutra_Cell2,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(0, 10), v_EncodedPdcpPdu)}));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    t_Watchdog.stop;
    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1, eutra_Cell2);//@sic R5s120629 sic@
    f_EUTRA_TestBody_Set(false);
    //test body ends
    ///////////////////////////////////////////////////////////////////////////////////
    //      Post Amble
    //////////////////////////////////////////////////////////////////////////////////
    // to configure SS to stop report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable );
    // to disable reception of SR in cell 2
    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell2, tsc_RbId_DRB1,cs_MAC_TestMode_NormalMode);
    // To deactivate test loop mode before RRC release
    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble (eutra_Cell2, E3_TESTMODE);  //@sic R5-103875 sic@
    //Remove the cell1, non camping
    //@sic R5-103847 taken care in postamble now sic@
  }; // end of 7.1.3.9
…
After change
…

  /*
   * @desc      REFERENCE TS 36.523-1 clause 7.1.3.9
   * @status    APPROVED (LTE)
   */
  function f_TC_7_1_3_9_EUTRA ( ) runs on EUTRA_PTC
  { /* L2/MAC
     * 7.1.3.9 : MAC reset-DL
     */
    const HarqProcessId_Type tsc_HARQ_Process := 3; // selected the highest allowed HARQ process in TDD config 0
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing2;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var SYSTEM_IND v_ReceivedSYSIND;
    var float v_HarqIndWindowSize;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    //@sic R5s110576 sic@
    var template (value) RadioBearerList_Type v_DRB :={cs_OneDRB_ConfigAM(tsc_DRB1, cds_PDCP_Config_DRB_AM_NoPDCP_StatusReport, //@sic R5s110687 sic@
                                                                          cs_RLC_Configuration_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)))};
    var template (value) DRB_ToAddModList v_DRB_ToAddModList := {cs_508_DRB_ToAddMod_Common(tsc_DRB1,
                                                                                            cds_PDCP_Config_DRB_AM_NoPDCP_StatusReport,//@sic R5s110687 sic@
                                                                                            cs_508_RLC_Config_DRB_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)),
                                                                                            cs_508_LogicalChannelConfig_DRB_AM)};
    var octetstring v_EncodedPdcpPdu;
    var octetstring v_EncodedRlcPdu;
    var SubFrameNumber_Type v_SubFrameRRCReconfigAck := 8; //@sic R5s170939 sic@
    timer t_Watchdog:=5.0; // General purpose wait timer
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    // Config Cell 1 for 1 AM DRB
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_SetCellPower(eutra_Cell1, -85);
    //@sic R5s100456, cell config moved above set attenuation  sic@
    // bring up cell 2 with T0 setting
    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);
    f_EUTRA_CellConfig_Def (eutra_Cell2);
    f_SS_ConfigDLHARQProcess_Subset(eutra_Cell1, -, cs_HarqProcessList_012);// To allow only first 3 HARQ processes for autonomous SS selection
    f_SS_ConfigDLHARQProcess_Subset(eutra_Cell2, -, cs_HarqProcessList_012);// To allow only first 3 HARQ processes for autonomous SS selection
    // Bring UE to  state 4
    f_EUTRA_Preamble ( eutra_Cell1,STATE2A_TESTLOOP_ModeA );
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    //@sic R5s110576 sic@
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell1, v_DRB);
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling(false, v_DRB_ToAddModList); //  p_PDCP_StatusReportRequired is false
    f_Delay(0.660); // SS is not stopped for TA transmission to prevent UE from entering DRX
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);
    // preconfig for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell2, tsc_C_RNTI_Def2);     /* @sic acc. to R5s100029 change 3 R5s120629 sic@ */
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    //@sic R5s150413 sic@
    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell2, cs_OneDRB_ConfigAM);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2); // @sic R5-101050 sic@
    //@siclog "Step 1" siclog@
    // To stop UL Grant Transmision
    //@sic R5s110201 sic@
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1,cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    //@sic R5s100852r2 MCC change sic@
    // to configure SS to report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // 38 bytes PDCP SDU +2 byte PDCP header
    // 2 bytes RLC Header +2 byte MAC Header =352 bits, nearest TB  size =376 => padding =24 bits
    // 1 byte padding header and 2 bits padding
    //@siclog "Step 1" siclog@
    // Sequence is changed to facilitate SS implementations.
    f_EUTRA_SetCellPower (eutra_Cell2, -79);
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 400);// schedule 400 ms in future
    // set CRC error as error mode
    f_SS_ConfigCRC_ErrorMode(eutra_Cell1, cs_DL_SCH_CRC_C_RNTI_Error(Erroneous), cs_TimingInfo(v_Timing));
    v_Timing.Subframe.Number := 0; // suitable for both FDD and TDD
    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 30);// schedule 20-30 ms in future
    //@siclog "Step 2" siclog@
    //@sic R5s120629 sic@
    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_L36)));
    v_EncodedRlcPdu   := f_RLC_AMD_1PDU_Encvalue(0, tsc_NoPoll, v_EncodedPdcpPdu);
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo(v_Timing), //cs_TimingInfo_Now,
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,
                                                                                             v_EncodedRlcPdu)),
                                          cs_HarqProcessAssignment_Specific(tsc_HARQ_Process)));
    // set CRC error as normal mode 60 ms after the scheduled transmission
    // so that 5 transmissions are performed.
    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);
    f_SS_ConfigCRC_ErrorMode(eutra_Cell1, cs_DL_SCH_CRC_C_RNTI_Error(Normal), cs_TimingInfo( v_Timing ));
    //@sic R5s120629 sic@
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1,cs_MAC_TestMode_NormalMode,-,cs_TimingInfo( v_Timing ));
    //@siclog "Step 5" siclog@
    // Tell the SS about the HO
    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverInit_REQ(eutra_Cell1, eutra_Cell2)); // @sic R5s100556 sic@
    v_Auth_Params := f_EUTRA_Security_Get();
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell2, v_Auth_Params, 0);
    f_EUTRA_Security_Set ( v_Auth_Params );
    f_EUTRA_SS_ConfigRachProcedure_HO(eutra_Cell2, cs_TimingInfo_Now, tsc_C_RNTI_Def2);
    f_EUTRA_CellInfo_SetC_RNTI (eutra_Cell2, tsc_C_RNTI_Def2);
    // stop periodic TA on source cell
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_None)); // Note: since the UE is configured for TA infinitive it does not matter that SS reconfiguration is not scheduled
    //@sic R5s150413 sic@
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell2, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    // configure UL grant configuration on target cell    @sic R5s100515 sic@
    f_EUTRA_ULGrantConfiguration_Common(eutra_Cell2, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def, cs_UplinkTimeAlignment_Synch_None, cs_UL_GrantConfig_Def);
    v_Timing2 := f_EUTRA_NB_TimingInfoAdd(v_Timing, 10);
    //@sic R5s100240 sic@
    f_EUTRA_508RRC_ConnReconfigHO_IntraLTE (eutra_Cell1,
                                            eutra_Cell2,
                                            cs_TimingInfo_SubFrameOffset( v_Timing2.SFN.Number, 4, v_Timing2.HSFN.Number )); // next sub frame suitable for both FDD and TDD @sic R5w160204r1 sic@
    //@siclog "Step 3" siclog@
    t_Watchdog.start;
    SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack ));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    t_Watchdog.stop;
    if (pc_ue_CategoryDL_M1)  //@sic R5s170939 sic@
    {
    v_SubFrameRRCReconfigAck := 0;
    v_Timing2.SFN.Number := ((v_Timing2.SFN.Number) + 1) mod 1024;
    }
    //@siclog "Step 6" siclog@
    // Check: does the UE transmit a RRCConnectionReconfigurationComplete message to confirm the reconfiguration of the radio resources?
    t_Watchdog.start;
    alt {
      [] SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) ) -> value v_SRB_COMMON_IND
        {
          v_Timing := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; //store TimeStamp when RRCConnectionReconfiguration has been received
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          t_Watchdog.stop;
        }
      []  SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack ))
        {
          repeat;
          // any NACKS received
        }
      //@sic R5s170939 sic@
      []  SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, cr_TimingInfo( v_Timing2.SFN.Number, v_SubFrameRRCReconfigAck ), ack )) // subframe 8 for ACK both for FDD and TDD configuration 1
        //note: retransmissions not allowed in this test case
        {
          repeat;
          // ACK  received for RRC connection reconfiguration
        }
    };
    //@siclog "Step 7" siclog@
    t_Watchdog.start (0.5); // watch dog timer to check 100ms as per prose which is less than 500ms, actual 100ms is checked based on time stamps
    alt {
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack )) -> value v_ReceivedSYSIND
        {
          //Check if the received HARQ Ind is within a window of 100 ms
          v_Timing2 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
          v_HarqIndWindowSize := f_EUTRA_SetTimerToleranceMin(eutra_Cell1, l2Timer, 0.1); //100 ms to wait safely no SR Rx
          if (f_EUTRA_SubFrameTimingDuration_Float(v_Timing, v_Timing2) < v_HarqIndWindowSize) {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7: UE is sending HARQ NACK in cell 1");
          }
        }
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell2, ?, nack )) -> value v_ReceivedSYSIND
        {
          //Check if the received HARQ Ind is within a window of 100 ms
          v_Timing2 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
          v_HarqIndWindowSize := f_EUTRA_SetTimerToleranceMin(eutra_Cell1, l2Timer, 0.1); //100 ms to wait safely no SR Rx
          if (f_EUTRA_SubFrameTimingDuration_Float(v_Timing, v_Timing2) < v_HarqIndWindowSize) {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7: UE is sending HARQ NACK in cell 2");
          }
        }
      [] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell2, ?, ack ))      // HARQ ACK for Status PDU sent by SS RLC for RLC reconfiguration complete
        {
          repeat;
        }
      []  t_Watchdog.timeout
        {
        }
    }
    // disable harq reporting in cell 2
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);
    // default grant config
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell2);
    f_EUTRA_SS_PdcpHandoverCtrl(eutra_Cell2, cas_PdcpHandoverComplete_REQ(eutra_Cell2)); // @sic R5s100556, R5s100560 sic@
    // configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell2);     /* @sic acc. to R5s100029 change 3 sic@ */
    //@sic R5s150992 sic@
    f_EUTRA_SS_SRBs_DRBs_Reset(eutra_Cell1);
    // to enable reception of SR in cell 2
    //@sic R5s110456 sic@
    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);
    //@siclog "Step 8" siclog@
    // SS guarantees NDI to be same as step2, as
    //"The initial value of the NDI shall be the same for all HARQ processes and cells".
    //@sic R5s120629 sic@
    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_L36_1)));
    v_EncodedRlcPdu   := f_RLC_AMD_1PDU_Encvalue(0, tsc_NoPoll, v_EncodedPdcpPdu);
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell2,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo_Now,
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,
                                                                                             v_EncodedRlcPdu)),
                                         cs_HarqProcessAssignment_Specific(tsc_HARQ_Process)));
    //@siclog "Step 9" siclog@
    t_Watchdog.start;
    SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell2, ?));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9: UE sent SR in cell 2");
    t_Watchdog.stop;
    //@siclog "Step 10" siclog@
    // Automatically taken care of
    //@siclog "Step 11" siclog@
    t_Watchdog.start;
    //@sic R5s120629 sic@
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (eutra_Cell2,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(0, 10), v_EncodedPdcpPdu)}));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    t_Watchdog.stop;
    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1, eutra_Cell2);//@sic R5s120629 sic@
    f_EUTRA_TestBody_Set(false);
    //test body ends
    ///////////////////////////////////////////////////////////////////////////////////
    //      Post Amble
    //////////////////////////////////////////////////////////////////////////////////
    // to configure SS to stop report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable );
    // to disable reception of SR in cell 2
    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell2, tsc_RbId_DRB1,cs_MAC_TestMode_NormalMode);
    // To deactivate test loop mode before RRC release
    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble (eutra_Cell2, E3_TESTMODE);  //@sic R5-103875 sic@
    //Remove the cell1, non camping
    //@sic R5-103847 taken care in postamble now sic@
  }; // end of 7.1.3.9
..
